

# COMMERCIAL-IN CONFIDENCE

SCIENCE RESEARCH COUNCIL  
RUTHERFORD & APPLETON LABORATORIES

COMPUTING DIVISION

D I S T R I B U T E D   C O M P U T I N G   N O T E   3 4 3

PERQ  
TECHNICAL NOTE 3  
Notes on Programming the PERQ

issued by  
R W Witty

28 January 1981

---

DISTRIBUTION:      F R A Hopgood  
                     R W Witty  
                     D A Duce  
                     W P Sharpe  
                     C J Prosser  
                     J Brown  
                     I D Benest  
                     A S Williams  
                     RL Support/Comp Facils/PERQ/General file

## 1. PERQ PERFORMANCE


```
for i: = 1 to 1000 do
for j: = 1 to 1000 do <statement>;
```

<u>&lt;statement&gt;</u>	<u>repetition</u>	<u>time</u>	<u>Comment</u>
Null	1 M	12 secs	
K: = K	1 M	15 secs	3 Q codes?
K: = K + 1	1 M	15 secs	4 Q codes?
K: = K + K	1 M	15 secs	"
K: = K * K	1 M	22 secs	"
K: = K div K	1 M	27 secs	"

## 2. FLOPPY FORMATTING

```
Run program TEST FLOPPY
DD? NO
Bttedo? YES
Interface? 5
Seed ? 56345 (a random number)
Format ? YES
Test ? Ynn
```


### 3. PLUG LAYOUT AT BOTTOM OF BACK



IMPORTANT: ALWAYS SWITCH OFF POWER BEFORE INSERTING OR REMOVING PLUG

Failure to do so can damage hardware!

### 4. INSTALLATION




1. Remove disk transit screw A.
2. Connect +5V, +12V plug B.
3. Connect 60 Hz power plug C.
4. Check all PC boards correctly seated.
5. Check all device plugs correctly connected.
6. Switch on.
7. Leave for 2 minutes whilst disk runs up to speed.
8. Should boot automatically.
9. If not, hit boot button on back of keyboard.
10. Check LED display for 999 (ok) else error code.

ALWAYS SWITCH OFF POWER BEFORE INSERTING/REMOVING ANY PLUG


(SRC PERQ : Slot #9, Monitor SN102, Frame (by power plug) SN110,  
CPU serial no 124).

## 5. FONT LAYOUT

Font layout is given in Module Screen, together with standard font associated routines, cursor control and window management.



Individual character definitions are stored as



See RASTOP program

Character patterns are stored in this way so that the width of the overall pattern area is the same (0 ... 767) as the screen. This is for Rasterop efficiency.

## 6. SEGMENTS AND POINTERS

Segment addresses are 16 bit segment numbers plus 16 bit offset within ( $2^{16} = 64K$ ) segment. Hence a 32 bit virtual address is constructed. The Pascal pointer is implemented as this 32 bit virtual address. Sometimes one must use a pointer (normal Pascal usage) and sometimes one must use a segment number, segment offset form (rasterop). The relationship between them is:

← 32 bit pointer or virtual address →

Segment number (16 bits)	Offset within segment (16 bits)
--------------------------	---------------------------------


In Pascal the two equivalent forms are handled by a variant record

```
Typptr = packed Record
  case boolean of
    true: (segm,offset:integer);
    false: (actptr :somepointertype);
  end;
```

eg lines 19-22 in RASTOP program.

Line 145 of RASTOP shows the conventional pointer form in use (actptr) whilst line 148 shows the segment, offset form of the same virtual address.

It is important to realise that the following are not equivalent



Due to the internal workings of the computer form (1) will produce the reverse ordering of the integers, and so cannot be used for virtual address equivalencing.

### 7. COMMAND WINDOW

The 'system' window is window zero.

### 8. RASTEROP

#### Alignment

Rasterop source and destinations must be quad word aligned because rasterop is 64 bit operation. Alignment is controlled by the extended NEW procedure. If the 2nd parameter is 4 then the data with the quad word aligned. Scan lines must be multiples of 4 words.

#### Efficiency

Rasterop is apparently more efficient if source and destination widths are multiples of four words and the same length. When rasteroping to/from screen it is apparently most efficient if both source and destination are 0 ... 767, ie 48 words wide. This is why font definitions are lines of 48 words (0 ... 767).

### 9. COMPILING LARGE PROGRAMS

Programs such as QUEEN require the magic words COM QUEEN/S:12. The default is 15. This must be reduced for large programs.

## 10. KNOWN COMPILER BUGS

- (1) for i:= 1 to 32767 will loop for ever.
- (2) end. needs a carriage return after the '.' else "unexpected eof" will occur.
- (3) write(file,char) does not work if file is 'var' parameter.
- (4) real numbers are not implemented. The compiler can handle them (in the latest version) but the microcode is still in preparation.

(Miles Barel has NPL validation suite. I left Unix Pascal VU tests).

## 11. KNOWN OPERATING SYSTEM BUGS

- (1) If one does too many edits then the editor runs out of memory and crashes (eg replace all spaces in big file).
- (2) DO NOT UNDER ANY CIRCUMSTANCES do a cntrl-C during the updating of a file by the editor (U) else one will have to hand patch the disk to recover.
- (3) SSetCursor is suspect if  $X \leq 10$ .

## 12. Z80

Don Selza is responsible for the Z80 code. It is all in assembler. There are 2 x 4K PROMs but the highest address used is #10466. Might be feasible to put CRing code in Z80. Z80 physically located in I/O board.

## 13. THE WSORT PROGRAM

The WSORT program is a modification of Bill Sharpe's DSORT1 program which runs on the Terak and is a simple animation of a SORT procedure. DSORT1 has been modified in two ways to make it run on the PERQ. Firstly, the Terak intrinsic 'GotoXY' has been rewritten using the PERQ graphics intrinsics, and secondly, the WSORT procedure has now been made to run in two separate windows by illustrating the use of the rather primitive PERQ window manager. Given below is a commentary on the various changes need to make DSORT1 into WSORT.

### Line 2

The import command is used to import type definitions and procedures into the user's program. It imports the module screen (note that this is spelt with 3 e's because of a silly typing error by Three Rivers) from the file screen. Module screen contains procedures for positioning the cursor, reading the cursor, setting the type of the cursor for creating and manipulating the windows and defining and changing fonts. See page 7 of the PERQ Operating System Programmer's Guide. Program WSORT uses 'set cursor' and 'create window', which are from module screen.

#### Lines 11-15

Terak intrinsic 'move cursor routine' called 'gotoXY' is directly equivalent to the PERQ procedure 'setcursor' and so in WSORT 'gotoXY' is just expanded to be a call of 'setcursor'. The complex parameters to 'setcursor' just cope with the finer resolution of the PERQ and cope with the fact that WSORT actually runs the DSORT program in two separate windows, hence the REPT MOD 2 component which determines which window is used.

#### Line 83

Due to a deficiency in the existing PERQ Pascal system, all programs must include the statements 'reset(input)', 'rewrite(output)' as their first statements.

#### Line 85

Character #14 is the form feed character. Using the standard I/O procedure 'write' to output the form feed character clears the screen. In the context of the standard output channel, form feed clears the screen, end of line character advances to the next line, and back space erases the previous character (see module screen).

#### Lines 90-94

The statements set up the extremities of the two windows in which the WSORT program will run. They are in absolute screen coordinates because the rather primitive screen manager only works in absolute screen coordinates.

#### Lines 96-103

This is a call to the screen manager which will create a new window which has a rectangular boarder with a black stripe at the top in which appears the title given on line 102. Its top left-hand corner will be WX1,WY1 and it will be WWD pixels wide and WXT pixels high. On line 97 the integer 1 is the window number. Window numbers are in the range 0-15. The WSORT program creates two slightly overlapping windows, one at line 96 and one at line 104.

#### Lines 113-126

DSORT1 program is repeated 4 times, twice in each window, creating a new window each time. After each window has been created the statement 'gotoXY (0,0) on line 124' causes the current cursor position to be moved to the origin, ie the top left-hand corner of the window. However, the screen manager is deficient in the sense that it does not automatically change the coordinate system when a new window is created. This means that anything that is written into a window must be absolute screen coordinates, hence the reason why the 'gotoXY (0,0)' which is relative to the window origin has to be transformed into absolute coordinates in the procedure 'gotoXY' so that 'setcursor' will work. See lines 11-15. Line 126, writing out the form feed character, will just clear the current window.

#### 14. THE RASTOP PROGRAM

The Rastop program is a tutorial exercise in using the Rasterop function directly from Pascal. Rasterop is used to draw an arbitrary pattern and move it, and is also used to access and draw the standard font.

##### Lines 3-6

Various PERQ operating system modules are imported.

##### Lines 9-13

A 2-dimensional array will be used to hold the pattern which is moved around the screen by Rasterop. It is COLBIT pixels wide x ROWBIT pixels deep. Note how COLBIT is defined as a multiple of 16 x COLWORD to ensure that it is word aligned. For greatest efficiency, COLBIT should be a multiple of 64 to ensure quad-word alignment. PATLEN is a number of bits needed to store a font. Note that PATLEN is 768 (the width of the screen in pixels) divided by 16 to give the number of words in a scanline and multiplied by 64 which is the maximum of lines in the standard font definition (see the definition of fontptr in module screen.

##### Lines 19-22

FPTR is pointer type which shows how the 32 bit virtual address, which is the Pascal pointer, is equivalent to the 16 bit segment number plus 16 bit offset.

##### Lines 39-40

'Createsegment' is one of the Memory Management procedures from module memory. Its function is to return to the user program the segment number of a new segment which can be used for the user's data. The segment number of the new segment is given by the first parameter SEGNO. The three input parameters are integers which define the initial size of the segment as a multiple of 256 words, the increment by which the segment may be expanded in terms of 256 words to a maximum specified by the third parameter. Line 39 creates a new segment whose initial size is 2 x 256 words which may be incremented 512 words at a time up to a maximum of 5 x 256 words. The new segment created by 'create segment' can have data put into it via pointers which are created by an extended version of the new procedure described in the Pascal Extensions manual. It takes three parameters in the extended version. The first is the segment number to be operated on, the second parameter controls alignment, and the third output parameter is the traditional parameter, ie it is the traditional pointer parameter to be returned by the new procedure. The value of 4 ensures that the source pointer will point to an area which is quad-word aligned which is essential for Rasterop operations.

##### Lines 43-46

This piece of code just sets up the pattern which will be moved about by Rasterop. It defines a rectangular area which is comprised of a set of vertical stripes.

#### Lines 48-60

These statements set up the parameters for Rasterop. Rasterop is described in the Pascal Extensions manual. The destination for all Rasterop operations to come on the screen must use the operating system segment which contains the screen buffer. This is called SCREENSEG. Line 55 shows the use of a PERQ Pascal extension 'makepointer' which is a way of creating an arbitrary pointer from a segment number, so SCREENSEG is the segment number, 0 is the offset within the segment, and BLKPTR is the type of the pointer which is returned. Line 54 sets up the number of words in a scanline. This is 48 when dealing with the screen.

#### Lines 63-74

This Rasterop draws the pattern at the top of the screen and the delay at line 77 causes it to stay there for a few seconds. Note the use on line 64 of a mnemonic to describe the Rasterop functions. These are imported from the module raster.

#### Lines 79-113

This piece of code causes the pattern to be moved from the top of the screen to the middle of the screen. It is essentially achieved by deleting the top line from the 50 which make up the pattern, and adding one to the bottom, which together with the delay at line 83 caused the pattern to move smoothly down the screen. Note how things are added and removed from the screen using the same function 'exclusive or'.

#### Lines 116-151

This section of the program is independent of the first section and shows how to manipulate the font. In places it contains alternative ways of doing things, hence certain things are commented out.

#### Line 116

This uses a procedure from the screen module to return a pointer to the current font. Note the use of the actptr discriminator to show that this is the 32 bit virtual memory conventional Pascal pointer operation.

#### Line 117

This is a call to the PERQ extended new operator to create a new data area of type patn (line 17) which is quad-word aligned (because the second parameter is 4) and which is in the 'default user data' segment because the segment number specified for the first parameter is 0.

#### Line 118

This operation copies the current font character definitions into newly created data area. Later on the code will show that the font can be painted on to the screen from either the system area of this newly created user area. This is shown in line 148 and 149. Line 130 is incorrect and should be disregarded. The Rasterop of line 119 paints the entire font up on the screen as two lines of text below which appears a lot of random dots. This is because the font actually needs less than 63 scanlines to define it all. From this operation it is clear that it is useful to define a font as being a set of rows which can be displayed across the screen as well as being efficient for Rasterop.

### Lines 134-151

This loop paints each individual character on the screen just below the entire font and shows how to access individual characters. Line 136 is a delay just to allow the character to be clearly seen. Lines 139 and 140 show how the height and width of the character are used in the Rasterop operation. Line 141 is the absolute screen coordinate for the X position of the character. Line 142 shows the absolute screen coordinate in the Y direction for the character. Note how the base value is subtracted from the Y coordinate. This ensures that the Y coordinate is the bottom left position of the character. Note one uses base not height, because the height could be variable for a character with a decending tail. Line 144 'destptr' is the screen (see line 55). Line 145 is the position of the character's left-most pixel in the scan line. Line 146 computes which row of scan lines the character starts. Note that this is computed by looking up which line (first or second in a standard font) of the font the character appears in, and then multiplying this by height or number of pixels per line. Line 147 is the number of words in a scanline. This is 48 for the standard font definitions, the same as the screen. Lines 148 and 149 are alternatives. Line 148 makes a pointer to the actual system area containing the standard font. However, it uses the equivalence mechanism to refer to the current font pointer the segment and offset method rather than the Pascal pointer. The magic number octal 404 is added to the offset value because 404 words is the number of words to implement the index array which preceeds the patterns in core. If line 149 was used instead of 148 the source rectangle for Rasterop would be the copy of the system font which was made by lines 117 and 118.

### 15. THE QUEENS PROGRAM

This program is a modification of Janet Malone's animation of the QUEENS program which runs on the Terak. The program essentially shows how the Terak intrinsics 'gotoXY', 'drawline' and 'drawblock' can be implemented by the PERQ intrinsics 'setcursor', 'drawline' and 'rasterop'.

### Lines 10-11

These two lines serve to import some definitions of segment numbers using the 'include' mechanism of the PERQ Pascal Compiler.

### Line 28

The Terak program used blocks as a type which were passed to the Terak screen handling procedures. PERQ Pascal is a little bit more rigorous and requires pointers to be passed which point to blocks which have been quad-word aligned (see line 404).

### Lines 29-33

This record type was defined in order to read in the photo file which gives the bit pattern to draw QUEEN. I never quite got the queen to come out right on the PERQ and I never found the reason for this, so this piece of code is suspect.

#### Lines 53-56

This shows how the Terak intrinsic 'gotoXY' for positioning the cursor is mimicked by the 'setcursor' routine. The 'setcursor' routine does not seem to work properly if the X coordinate is less than 10. I suspect this is just a bug as  $X < 10$  is around one character width. X and Y are scaled and 'setcursor' works in absolute screen coordinates, not characters.

#### Lines 75-109

Terak intrinsic 'drawblock' is mimicked by Rasterop. Lines 89-96 essentially convert the 'drawblock' mode to its equivalent Rasterop function. The actual conversion in these few lines is probably not quite accurate as I could not remember all the exact values of 'drawblock' and Janet had very kindly used integers rather than mnemonics in her program. Line 95 shows the use of the PERQ Pascal extension to the case statement. For details of the Rasterop function see above. However, note line 104 making a pointer to the system screen buffer, and line 107 where the number of pixels in the scan line which, supplied by 'drawblock', is converted to words per scan line. The Terak procedure 'drawblock' does not take any cognisance of quad-word alignment and so the source parameter must be manipulated outside the 'draw block' procedure to ensure it is quad-word aligned. An alternative to this would be to copy the contents of the source to some quad-word aligned buffer before doing Rasterop, but this would seem to be a little slow.

#### Lines 114-118

The Terak delay intrinsic 'throttle' has been simulated by just wasting time.

#### Lines 122-145

The Terak intrinsic 'drawline' (converted to 'drawaline' here to avoid a name clash) is quite simply re-implemented using the PERQ line mechanism. This is defined in the line draw module.

#### Lines 187-198

The Terak routine 'fillchar' has no equivalent in PERQ Pascal and so lines 195-198 simulate its effects. These are a little tricky because the version on the PERQ comes out as the inverse of the Terak, ie everything that is black on the Terak is white on a PERQ. Note how lines 195-198 operate on 32 x 32 arrays to ensure word alignment.

#### Lines 208-211

These two commands clear the screen and set the Terak screen buffer to be 'screen'. In the PERQ version of the standard write command to clear the screen and the system screen buffer is always used rather than the user defined screen buffer, hence the use of 'makeptr (screenseg)' in the Rasterop operations.

Lines 229-235

This little bit of jiggery pokery reads in the Terak photo file which defines the bitmap of a Queen. I never did quite get this queen to come out properly on the PERQ screens. I am sure there is something wrong with this bit of code. The queen always came out with what looked like the left third of it missing which seemed to suggest some kind of word alignment problem and I never solved it. I am sure it is some stupid error somewhere.

Lines 400-404

Reset and rewrite must appear in every program. Line 401 clears the screen, and line 404 ensures that the various buffers used in Rasterop operations are quad-word aligned and appear in the user's default data segment segment number 0.

APPENDIX 1

<u>NAME</u>	<u>EXTENSION</u>	<u>DEPARTMENT</u>	<u>ROOM</u>
Barel, Miles	37	Software	302
Broadley, Bill	40	Hardware	307
Cercone, Richard	27	Drafting	205B
Crenner, Jim	22	Test Office	107
Eckenrode, Chuck	23	Test Area	106
Fredkin, Edward	34	C. E. O.	209
Gasbarro, Jim	41	Hardware	306
Gilmour, Tom	27	Drafting	205B
Glass, Bill	38	Software	301
Grandey, Connie	21	Assembly Area	104
Hahn, Andy	20	Shipping/Receiving	103
Harrington, John	25	Accounting	204
Heller-Hendrick, Jean	21	Assembly Office	105
Heller, Sandy	21	Assembly Area	104
Horner, Jake	21	Assembly Area	104
Horner, Michelle	20	Stockroom	103
Howell, Jeff	24	Manufacturing	204
Janets, Lynne	26	Bookkeeping	203
Letterle, Bruce	29	Personnel	201
Letterle, Karen	0	Receptionist	
Manko, Jack	30	Production	206
Marks, Diane	33	Secretary	208
Myers, Brad	37	Software	302
Newbury, Paul	35	President	210
Peckar, Jeff	26	Software Support	203
Reddy, Pradeep	41	Hardware	306
Reid, Jill	20	Stockroom	103
Rose, John	40	Hardware	307
Rosen, Brian	31	Vice President	211
Scelza, Don	36	Software	303
Schultz, Dorothy	21	Assembly Area	104
Software Lab	39		308
Steele, Guy			
Stewart, Rege	23	Test Area	106
Strait, John	36	Software	303
Vavra, Terry	23	Test Area	106
Visnich, Donna	27	Drafting	205B
Williamson, Theresa	21	Assembly Area	104

```

00001:PROGRAM wsort(input,output);
00002:imports screen from screen; {ed}
00003:const max=11; firstrow=1; cx=35; cy=23; pfac=5; {ed}
00004:var n,srccol,dstcol,srcpos,dstpos,
00005: win,
00006: i,j,count,searchcount,a,b,total,rept :integer;
00007: col :array[1..3] of integer;
00008: list :array[1..max] of integer;
00009: wx1,wy1,wwd,wht :array[1..2] of integer;
00010: change : boolean;
00011:procedure gotoxy(X,Y:integer); {ed}
00012:begin
00013: ssetcursor(X*10+wx1[(rept mod 2)+1]+30,
00014: Y*13+wy1[(rept mod 2)+1]+50);
00015:end;
00016:
00017:
00018:procedure writetotal;
00019:begin
00020: total:=total+1;
00021: gotoxy(cx+24,cy);
00022: write(total:2)
00023:end;
00024:
00025:procedure delay(m : integer);
00026:var i,j : integer;
00027:begin
00028: for i:=1 to m do j:=0 ;
00029:end;
00030:
00031:
00032:procedure flash(column, row : integer) ; {draw attention to list item}
00033:var x,y : integer;
00034:
00035:begin
00036: x:=col[column]-4; y:=firstrow+(row-1)*2;
00037: gotoxy(x,y); write(chr(7),'>>>'); gotoxy(x,y);
00038: delay(2500*pfac);
00039: write(' ')
00040:end;
00041:
00042:procedure shift(n,srccol,srcpos,dstcol,dstpos : integer);
00043:var i,m,x1,x2,y1,y2 :integer;
00044:begin
00045: m:=20; y1:=firstrow+(srcpos-1)*2;
00046: y2:=firstrow+(dstpos-1)*2;
00047: if dstcol>=srccol then
00048: begin
00049: x1:=col[srccol];
00050: x2:=col[dstcol];
00051: for i:=1 to n do
00052: begin
00053: gotoxy(x1,y1); write(' ');
00054: gotoxy(x2,y2); write('*');
00055: delay(1*pfac);
00056: x1:=x1+1; x2:=x2+1
00057: end
00058: end
00059: else
00060: begin
00061: x1:=col[srccol]+n-1;
00062: x2:=col[dstcol]+n-1;
00063: for i:=1 to n do
00064:

```

```

00004:          begin
00065:              gotoxy(x1,y1); write(' ');
00066:              gotoxy(x2,y2); write('*');
00067:              x1:=x1-1; x2:=x2-1;
00068:              delay(1*pfac)
00069:          end;
00070:          gotoxy(x2,y2)
00071:
00072:      end
00073: end;
00074:
00075: function next(n : integer) : integer;
00076: {returns next nonnull position in list AFTER n}
00077: begin
00078:     next:=n+1;
00079:     if n+1 <=max then if list[n+1]=0 then next:=next(n+1);
00080: end;
00081:
00082: begin
00083:     reset(input); rewrite(output); {ed}
00084:
00085:     write(chr(#14));
00086:
00087:
00088:
00089:
00090:     wx1[1]:=0;          wy1[1]:=0;
00091:     wwd[1]:=700;       wht[1]:=500;
00092:
00093:     wx1[2]:=50;        wy1[2]:=320;
00094:     wwd[2]:=700;       wht[2]:=500;
00095:
00096:     CreateWindow(
00097:         1,
00098:         wx1[1],
00099:         wy1[1],
00100:         wwd[1],
00101:         wht[1],
00102:         'win one'
00103:     );
00104:     CreateWindow(
00105:         2,
00106:         wx1[2],
00107:         wy1[2],
00108:         wwd[2],
00109:         wht[2],
00110:         'win two'
00111:     );
00112:
00113:     for rept := 1 to 4 do
00114:         begin
00115:             {ChangeWindow((rept mod 2)+1);}
00116:             win:= (rept mod 2) + 1;
00117:             CreateWindow(win,
00118:                 wx1[win],
00119:                 wy1[win],
00120:                 wwd[win],
00121:                 wht[win],
00122:                 'window');
00123:
00124:             gotoxy(0,0);
00125:
00126:             write(chr(#14)); {ed}
00127:             col[1]:=10;
00128:             col[2]:=30;
00129:             col[3]:=50;
00130:

```

```

00130:      list[1] :=3,
00131:      list[2] :=4;
00132:      list[3] :=2;
00133:      list[4] :=5;
00134:      list[5] :=1;
00135:      list[6] :=4;
00136:      list[7] :=8;
00137:      list[8] :=6;
00138:      list[9] :=9;
00139:      list[10] :=10;
00140:      list[11] :=7;
00141:      for i:=1 to max do
00142:          shift(list[i],1,i,1,i);
00143:      total:=0; gotoxy(cx,cy); write('Number of comparisons = ');
00144:      count:=max {must deal with all one at a time};
00145:      repeat
00146:      begin
00147:          {find shortest}
00148:          searchcount:=count-1 ;
00149:          a:= next(0) {first non null list entry};
00150:          b:=a;
00151:          shift(list[a],1,a,2,a) {pick out};
00152:          while searchcount<>0 do
00153:          begin
00154:              delay(1000*pfac);
00155:              b:=next(b) {next non null entry};
00156:              shift(list[b],1,b,2,b) {pick out};
00157:              delay(1000*pfac);
00158:              change:=list[b] < list[a];
00159:              if change then
00160:                  begin shift(list[a],2,a,1,a){putback a};
00161:                      a:=b
00162:                  end
00163:                  else- shift(list[b],2,b,1,b){putback b} ;
00164:              searchcount:=searchcount-1;
00165:              writetotal
00166:          end;
00167:          delay(1000*pfac);
00168:
00169:          {move shortest to end of sorted list}
00170:          flash(2,a);
00171:          shift(list[a],2,a,3,max-count+1);
00172:          list[a]:=0 {delete from list};
00173:          count:=count-1;
00174:          delay(2500*pfac)
00175:      end;
00176:      until count=0;
00177:
00178:
00179:
00180:
00181:      end {of outer window changing loop} ;
00182:
00183:      end.

```

OK,

```

00001:program rastop(input,output);
00002:
00003:imports memory    from memory;
00004:imports linedraw  from linedraw;
00005:imports raster    from raster;
00006:imports screen    from screen;
00007:
00008:
00009:const
00010:  rowbit=50;
00011:  colwrd=20;
00012:  colbit=colwrd*16;
00013:  patlen=(768 div 16)*64;
00014:
00015:type blk = packed array[1..rowbit,1..colbit] of boolean;
00016:  blkptr = blk;
00017:  patn   = array[0..patlen] of integer;
00018:  patnptr = ^^patn;
00019:  fptr = packed record case boolean of
00020:    true:(segm,offs:integer);
00021:    false:(actptr:fontptr);
00022:  end;
00023: var
00024:  segno,
00025:  i,j,
00026:  func,wid,hgt,
00027:  destx,desty,destl,
00028:  sorcx,sorey,sorcl :integer;
00029:  destptr,sorcptr   :blkptr;
00030:  cfptr   : fptr;
00031:  {cfptr  : fontptr;}
00032:  patptr  : patnptr;
00033:
00034:begin
00035:  reset(input); rewrite(output);
00036:
00037:  write(chr(#14));
00038:
00039:  CreateSegment(segno,2,2,5);
00040:  new(segno,4,sorcptr);
00041:
00042:
00043:  for i:=1 to rowbit do
00044:    for j:=1 to colbit do
00045:      if odd(j) then sorcptr^[i,j]:=true
00046:        else sorcptr^[i,j]:=false;
00047:
00048:  func:=0;
00049:  wid :=colbit;
00050:  hgt :=rowbit;
00051:
00052:  destx:=100;
00053:  desty:=1;
00054:  destl:=768 div 16;
00055:  destptr:=makeptr(screenseg,0,blkptr);
00056:
00057:  sorcx:=0;
00058:  sorey:=0;
00059:  sorcl:=colwrd;
00060:  {sorcptr}
00061:
00062:
00063:  RASTEROP
00064:    (RXor,

```

```

00065:         wid,
00066:         hgt,
00067:         destx,
00068:         desty,
00069:         destl,
00070:         destptr,
00071:         sorcx,
00072:         sorcy,
00073:         sorcl,
00074:         soreptr);
00075:
00076:
00077:     for i:=1 to 30000 do j:=j;
00078:
00079:     for desty:=1 to 500 do
00080:         begin
00081:
00082:
00083:         for i:=1 to 500 do j:=j;
00084:
00085:         RASTEROP
00086:             (RXor,
00087:             wid,
00088:             1,
00089:             destx,
00090:             desty,
00091:             destl,
00092:             destptr,
00093:             sorcx,
00094:             sorcy,
00095:             sorcl,
00096:             soreptr);
00097:
00098:         RASTEROP
00099:             (RXor,
00100:             wid,
00101:             1,
00102:             destx,
00103:             desty+rowbit,
00104:             destl,
00105:             destptr,
00106:             sorcx,
00107:             sorcy,
00108:             sorcl,
00109:             soreptr);
00110:
00111:
00112:
00113:         end;
00114:
00115:
00116:     cfptr.actptr:=GetFont;
00117:     new(0,4,patptr);
00118:     for i:=0 to patlen do patptr^[i]:=cfptr.actptr^.pat[i];
00119:     RASTEROP
00120:         (RRpl,
00121:         768,
00122:         63,
00123:         0,
00124:         desty,
00125:         destl,
00126:         destptr,
00127:         0,
00128:         0,
00129:         48,
00130:         {makeptr(0,cfptr.actptr^.pat,LinePtr)}

```

```
00131:         patptr
00132:         );
00133:
00134:     for i:=0 to #177 do
00135:         begin
00136:             for j:=1 to 30000 do destx:=destx;
00137:             RASTEROP
00138:                 (RRpl,
00139:                 cfptr.actptr^^.index[i].width,
00140:                 cfptr.actptr^^.height,
00141:                 350,
00142:                 (desty+100)-cfptr.actptr^^.base,
00143:                 destl,
00144:                 destptr,
00145:                 cfptr.actptr^^.index[i].offset,
00146:                 cfptr.actptr^^.index[i].line*cfptr.actptr^^.height,
00147:                 48,
00148:                 makeptr(cfptr.segm,cfptr.offt+#404,fontptr)
00149:                 {patptr}
00150:                 );
00151:         end;
00152:
00153:
154:end.
```

OK,

```

00001:PROGRAM QUEENS;
00002:
00003:
00004:imports linedraw from linedraw;
00005:imports screen from screen;
00006:
00007:
00008:
00009:
00010:const
00011:{$I Segnumbers}
00012:
00013:
00014:TYPE TSCREEN = RECORD
00015:     CASE INTEGER OF
00016:         1 : (BITS : PACKED ARRAY[0..239] OF
00017:             PACKED ARRAY[0..319] OF BOOLEAN);
00018:         2 : (CHRS : PACKED ARRAY[0..9599] OF CHAR)
00019:     END;
00020: BLOCK = RECORD
00021:     CASE INTEGER OF
00022:         1 : (BITS : PACKED ARRAY[0..63] OF
00023:             PACKED ARRAY[0..63] OF BOOLEAN);
00024:         2 : (CHRS : PACKED ARRAY[0..512] OF CHAR)
00025:     END;
00026:
00027:
00028: blkptr = ^^block;
00029: queenin = record
00030:     case integer of
00031:         1:(bits:packed array[0..31,0..31] of boolean) ;
00032:         2:(chrs:packed array[0..127] of char)
00033:     end;
00034:
00035:
00036:VAR DONE,SAFE : BOOLEAN;
00037:     I,N : INTEGER;
00038:     A : ARRAY[0..7] OF BOOLEAN; { rows }
00039:     B : ARRAY[0..14] OF BOOLEAN; { / diagonals }
00040:     C : ARRAY[-7..7] OF BOOLEAN; { \ diagonals }
00041:     SOL : ARRAY[0..7] OF 0..7; { row posn. for each column }
00042:     SCREEN : TSCREEN;
00043:     BLACK,
00044:     CHECK,QUEEN : blkptr;
00045:     ANS : CHAR;
00046:     quin : queenin;
00047:
00048:
00049:
00050:
00051:
00052:
00053:procedure GOTOXY(x,y : integer);
00054: begin
00055:     ssetcursor(x*10+10,y*13+500);
00056: end;
00057:
00058:
00059:procedure FILLCHAR({blk:block.chrs;}
00060:     a,b:integer);
00061:begin
00062:end;
00063:

```

```

00064:
00065:procedure UNITWRITE(a:integer; var scrn:tscreen; b:integer);
00066:begin
00067:end;
00068:
00069:
00070:
00071:
00072:
00073:
00074:
00075:  PROCEDURE DRAWBLOCK(VAR SOURCE : blkptr;
00076:                      SRCROW,
00077:                      SRCX,
00078:                      SRCY : INTEGER;
00079:                      VAR  DEST : TSCREEN;
00080:                      DSTROW,
00081:                      DSTX,
00082:                      DSTY,
00083:                      CNTX,
00084:                      CNTY,
00085:                      MODE : INTEGER);
00086:
00087:  var
00088:    func:integer;
00089:  begin
00090:    case mode of
00091:      0:func:=0;
00092:      1:func:=1;
00093:      2:func:=6;
00094:      3:func:=4;
00095:    otherwise:func:=mode;
00096:  end;
00097:
00098:    RASTEROP(func,
00099:             cntx,
00100:             cnty,
00101:             dstx,
00102:             dsty,
00103:             48,
00104:             makeptr(screenseg,0,Lineptr),
00105:             srcx,
00106:             srcy,
00107:             (srcrow div 16),
00108:             source);
00109:  end;
00110:
00111:
00112:
00113:
00114:  PROCEDURE THROTTLE(TICKS : INTEGER);
00115:  var i,j:integer;
00116:  begin
00117:    for i:=1 to ticks*900 do j:=0 ;
00118:  end;
00119:
00120:
00121:
00122:  PROCEDURE DRAWALINE(VAR RANGE : INTEGER;
00123:                     VAR SCREEN : TSCREEN;
00124:                     ROWWIDTH,
00125:                     XSTART,
00126:                     YSTART,
00127:                     DELTAX,
00128:                     DELTAY,
00129:                     PENSTATE : INTEGER);

```

```

00130:
00131:   var ltyp:Linestyle;
00132:   begin
00133:     case penstate of
00134:       1:ltyp:=Drawline;
00135:       2:ltyp:=EraseLine;
00136:     otherwise:
00137:       ltyp:=XorLine;
00138:     end;
00139:     Line(ltyp,
00140:       xstart,
00141:       ystart,
00142:       xstart+deltax,
00143:       ystart+deltay,
00144:       makeptr(screenseg,0,LinePtr));
00145:   end;
00146:
00147:
00148:
00149:
00150:
00151:   PROCEDURE INTRO;
00152: BEGIN
00153:   WRITELN(
00154: '          THE QUEENS PROBLEM');
00155:   WRITELN(
00156: '          *****');
00157:   WRITELN(
00158: '    Given an N x N checkered board, the aim is to place N queens on the');
00159:   WRITELN(
00160: 'board so that no queen may take another (i.e. none are on the same');
00161:   WRITELN(
00162: 'diagonal, horizontal or vertical line).');
00163:   WRITELN(
00164: 'This may be solved recursively - for each column of the board, the');
00165:   WRITELN(
00166: 'queen is moved down each square until a position is found where she is');
00167:   WRITELN(
00168: 'safe. If there is no possible square in a column then we return to the');
00169:   WRITELN(
00170: 'previous column and choose a new square - this is known as ');
00171:   WRITELN(
00172: 'backtracking'.');
00173:   WRITELN(
00174: '    When the last queen is placed safely on the last column then this is');
00175:   WRITELN(
00176: 'a possible solution, but if all combinations of squares have been tried');
00177:   WRITELN(
00178: 'and failed then there is no solution.').
00179: END; { intro }
00180: PROCEDURE INITIALISE;
00181: TYPE BLOK = FILE OF queenin;
00182: VAR DUMMY,
00183:     I,j,X,Y : INTEGER;
00184:     CROWN : BLOK;
00185:     PROCEDURE SETCHECK; { creates a block of 24 x 24 checkered dots }
00186:     BEGIN { and sets up blank square }
00187:       {
00188:         FOR I := 0 TO 11 DO
00189:         BEGIN
00190:           FILLCHAR(CHECK.CHRS[I*8],4,85);   sets '...'
00191:           FILLCHAR(CHECK.CHRS[I*8+4],4,170) sets '...'
00192:         END;
00193:         FILLCHAR(BLACK,128,0)
00194:       }
00195:     for i:=0 to 31 do for j:=0 to 31 do

```

```

00196:         if odd(j) then check^^.bits[i,j]:=odd(i)
00197:                 else check^^.bits[i,j]:=not odd(i);
00198:         for i:=0 to 31 do for j:= 0 to 31 do black^^.bits[i,j]:=false;
00199:     END; {SETCHECK}
00200: BEGIN { initialise }
00201:     DONE := FALSE;
00202: {initialise arrays}
00203:     FOR I := 0 TO N-1 DO SOL[I] := 0;
00204:     FOR I := 0 TO N-1 DO A[I] := TRUE;
00205:     FOR I := 0 TO 2*(N-1) DO B[I] := TRUE;
00206:     FOR I := -(N-1) TO N-1 DO C[I] := TRUE;
00207: {draw board}
00208:     {
00209:     FILLCHAR(SCREEN,9600,0);
00210:     UNITWRITE(3,SCREEN,63);
00211:     }
00212:     SETCHECK;
00213:     FOR Y := 0 TO N-1 DO
00214:     BEGIN
00215:         IF ODD(Y) THEN X := 1
00216:             ELSE X := 0;
00217:         REPEAT
00218:             DRAWBLOCK(CHECK,32,0,0,SCREEN,320,24*X+125,24*Y+36,24,24,0);
00219:             X := X+2
00220:         UNTIL X > N-1
00221:     END;
00222:     FOR I := 0 TO 2 DO
00223:     BEGIN
00224:         DRAWALINE(DUMMY,SCREEN,20,122,33+I,24*N+5,0,1);
00225:         DRAWALINE(DUMMY,SCREEN,20,122,36+24*N+I,24*N+5,0,1);
00226:         DRAWALINE(DUMMY,SCREEN,20,122+I,36,0,24*N,1);
00227:         DRAWALINE(DUMMY,SCREEN,20,125+24*N+I,36,0,24*N,1)
00228:     END;
00229: { get queen character }
00230:     RESET(CROWN,'rob.QU.FOTO');
00231:     quin := CROWN^^;
00232:     CLOSE(CROWN);
00233:     for i:=0 to 31 do
00234:         for j:=0 to 31 do
00235:             queen^^.bits[i,j]:=quin.bits[i,j];
00236:
00237: { draw N queens above board }
00238:     FOR I := 0 TO N-1 DO
00239:         DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,I*24+125,9,24,24,0);
00240:     THROTTLE(25);
00241: { write explanation }
00242:     GOTOXY(0,14);
00243:     WRITELN('LINES - SHOW WHICH QUEEN');
00244:     WRITELN(' ':8,'WOULD TAKE THE NEW');
00245:     WRITELN(' ':8,'QUEEN IF HER SQUARE');
00246:     WRITELN(' ':8,'IS NOT SAFE. ');
00247:     WRITELN('**BACKTRACK** - THERE IS');
00248:     WRITELN(' ':8,'NO SAFE SQUARE FOR');
00249:     WRITELN(' ':8,'THIS COLUMN, SO WE');
00250:     WRITELN(' ':8,'MOVE THE QUEEN IN');
00251:     WRITELN(' ':8,'THE PREVIOUS COLUMN. ');
00252:
00253:
00254:
00255:
00256:     END; { initialise }
00257: PROCEDURE TRYCOL(J:INTEGER);
00258: VAR I,K,X,Y:INTEGER;
00259: PROCEDURE LINE; { draws lines showing if queen can be taken
00260:                 if placed on this square }
00261: VAR DUMMY,DX,DY,X,Y : INTEGER;

```

```

00262: BEGIN { line }
00263:   IF NOT(A[I]) { rows }
00264:   THEN BEGIN
00265:     DRAWALINE(DUMMY, SCREEN, 20, 125, I*24+48, 24*N, 0, 3);
00266:     THROTTLE(25);
00267:     DRAWALINE(DUMMY, SCREEN, 20, 125, I*24+48, 24*N, 0, 3)
00268:   END;
00269:   IF NOT(B[I+J]) { / diagonals }
00270:   THEN BEGIN
00271:     DX := (N - (ABS(I+J-N+1)))*24;
00272:     DY := -DX;
00273:     IF I+J < N
00274:     THEN BEGIN
00275:       X := 125;
00276:       Y := (I+J+1)*24+36
00277:     END
00278:     ELSE BEGIN
00279:       X := (I+J-N+1)*24+125;
00280:       Y := N*24+36
00281:     END;
00282:     DRAWALINE(DUMMY, SCREEN, 20, X, Y, DX, DY, 3);
00283:     THROTTLE(25);
00284:     DRAWALINE(DUMMY, SCREEN, 20, X, Y, DX, DY, 3)
00285:   END;
00286:   IF NOT(C[I-J]) { \ diagonals }
00287:   THEN BEGIN
00288:     DX := (N - (ABS(I-J)))*24;
00289:     DY := DX;
00290:     IF I-J < 0
00291:     THEN BEGIN
00292:       X := (J-I)*24+125;
00293:       Y := 36
00294:     END
00295:     ELSE BEGIN
00296:       X := 125;
00297:       Y := (I-J)*24+36
00298:     END;
00299:     DRAWALINE(DUMMY, SCREEN, 20, X, Y, DX, DY, 3);
00300:     THROTTLE(25);
00301:     DRAWALINE(DUMMY, SCREEN, 20, X, Y, DX, DY, 3)
00302:   END
00303: END; { line }
00304: PROCEDURE WIPEOUT(J, I: INTEGER); { removes queen character from square }
00305: BEGIN
00306:   IF ODD(I+J) THEN { black square }
00307:     DRAWBLOCK(BLACK, 32, 0, 0, SCREEN, 320, 24*J+125, 24*I+36, 24, 24, 0)
00308:   ELSE { grey square }
00309:     DRAWBLOCK(CHECK, 32, 0, 0, SCREEN, 320, 24*J+125, 24*I+36, 24, 24, 0)
00310:   END; { wipeout }
00311: PROCEDURE SETQUEEN; { sets row and diagonals as covered }
00312: BEGIN
00313:   A[I]:=FALSE; B[I+J]:=FALSE; C[I-J]:=FALSE
00314: END; { setqueen }
00315: PROCEDURE REMOVEQUEEN; { releases row and diagonals, signals
00316:   backtracking has occurred }
00317: VAR K : INTEGER;
00318: BEGIN
00319:   A[I]:=TRUE; B[I+J]:=TRUE; C[I-J]:=TRUE;
00320:   GOTOXY(0, J+2);
00321:   WRITE('***BACKTRACK**');
00322:   FOR K := 1 TO 12 DO
00323:     BEGIN
00324:       DRAWBLOCK(QUEEN, 32, 0, 0, SCREEN, 320, 24*J+125, 24*I+36, 24, 24, 2);
00325:       THROTTLE(6)
00326:     END;
00327:     GOTOXY(0, J+2);

```

```

00328:      {WRITE(CHR(29))}
00329:      for k:=1 to 13 do write(' ');
00330:
00331: END; { removequeen }
00332: PROCEDURE LASTCOLUMN; { a solution has been found. If another
00333:      solution is not required then the flag DONE is set }
00334: BEGIN
00335:     GOTOXY(0,9);
00336:     WRITELN('WOULD YOU LIKE ANOTHER');
00337:     WRITELN('SOLUTION FOR THIS BOARD? (Y/N)');
00338:     READLN(ANS);
00339:     GOTOXY(0,9);
00340:     WRITELN(CHR(29));
00341:     WRITELN(CHR(29));
00342:     WRITELN(CHR(29));
00343:     IF ANS <> 'Y' THEN DONE := TRUE
00344:         ELSE { remove queen }
00345:             A[I]:=TRUE; B[I+J]:=TRUE; C[I-J]:=TRUE
00346: END; { lastcolumn }
00347: PROCEDURE MOVEDOWN; { moves queen down one square of board }
00348: BEGIN
00349:     WIPEOUT(J,I);
00350:     DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,24*J+125,24*I+48,24,24,3);
00351:     THROTTLE(6);
00352:     WIPEOUT(J,I);
00353:     WIPEOUT(J,I+1);
00354:     DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,24*J+125,24*(I+1)+36,24,24,3);
00355:     THROTTLE(6)
00356: END; { movedown }
00357: BEGIN { trycol }
00358:     { remove queen from side of board and place on first square }
00359:     THROTTLE(25);
00360:     DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,J*24+125,9,24,24,2);
00361:     DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,24*J+125,36,24,24,3);
00362:     { write next line of solution }
00363:     GOTOXY(0,J+1);
00364:     WRITELN('(' ,J+1:1,' , )');
00365:     I := 0;
00366:     REPEAT
00367:         { fill in row of solution }
00368:         GOTOXY(3,J+1);
00369:         WRITE(I+1:1);
00370:         SAFE:= A[I] AND B[I+J] AND C[I-J];
00371:         IF SAFE
00372:             THEN BEGIN
00373:                 SETQUEEN;
00374:                 SOL[J]:=I;
00375:                 IF J < N-1
00376:                     THEN BEGIN
00377:                         TRYCOL(J+1);
00378:                         IF NOT DONE THEN REMOVEQUEEN
00379:                     END
00380:                     ELSE LASTCOLUMN
00381:                 END
00382:             ELSE LINE;
00383:         { move queen down 1 square, or replace above board }
00384:         IF (I < N-1) AND NOT DONE
00385:             THEN MOVEDOWN
00386:         ELSE IF (I = N-1) AND NOT DONE
00387:             THEN BEGIN
00388:                 WIPEOUT(J,I);
00389:                 DRAWBLOCK(QUEEN,32,0,0,SCREEN,320,J*24+125,
00390:                     9,24,24,0);
00391:                 { clear solution line }
00392:                 GOTOXY(0,J+1);
00393:                 WRITE(CHR(29));

```

```

00394:                                THROTTLE(25)
00395:                                END;
00396:                                I := I + 1
00397:    UNTIL (I=N) OR DONE
00398:END; { trycol }
00399:BEGIN { main }
00400:    reset(input); rewrite(output);
00401:    write(chr(#14));
00402:    gotoxy(0,0);
00403:
00404:    new(0,4,queen); new(0,4,check); new(0,4,black);
00405:
00406:    INTRO;
00407:    REPEAT
00408:        WRITELN('PLEASE TYPE IN N, THE SIZE');
00409:        WRITELN('OF THE BOARD (BETWEEN 1 AND 8)');
00410:        READLN(N);
00411:        IF (N<1) OR (N>8)
00412:        THEN WRITELN('BOARD SIZE SHOULD BE BETWEEN 1 AND 10')
00413:        ELSE BEGIN
00414:            {PAGE(OUTPUT);} write(chr(#14));
00415:            INITIALISE;
00416:            GOTOXY(0,0);
00417:            WRITELN('SOLUTION FOR ',N,' BY ',N,' BOARD :');
00418:            TRYCOL(0);
00419:            IF NOT DONE THEN WRITELN('NO SOLUTION')
00420:        END;
00421:        GOTOXY(0,30);
00422:        WRITELN('DO YOU WANT TO CONTINUE? (Y/N)');
00423:        READLN(ANS)
00424:    UNTIL ANS <> 'Y'
00425:END. { main }

```

OK,